STATS 261 SAS LAB FOUR, February 4, 2009

Lab Four: PROC LOGISTIC

Lab Objectives

After today’s lab you should be able to:

1. Use PROC LOGISTIC for multivariate logistic regression.

2. Interpret output from PROC LOGISTIC.

3. Understand how to deal with continuous and categorical predictors in PROC LOGISTIC.

4. Use and understand the “units” statement in PROC LOGISTIC for generating meaningful odds ratios from continuous predictors.

5. Use the class statement for categorical predictors to generate odds ratios for comparisons against a reference group.

6. Test whether a predictor is linear in the logit.

7. Get predicted probabilities from a logistic model.

8. Get predicted probabilities for new observations.

9. Run a MACRO that someone else has written.

10. Test for confounding.

11. Test for interaction.

12. Use a BY statement for stratifying.

 LAB EXERCISE STEPS:

Follow along with the computer in front…

1. Goto the class website: www.stanford.edu/~kcobb/coruses/hrp261--> Download LAB 3 Data. Right click to save data on the desktop as an excel file: psa.xls. Also download Logit Plot Macro on your desktop.
2. Open SAS. Import the data into SAS using point-and-click: 

a. Goto: File--> Import Data-->to open Import Wizard

b. Select Microsoft Excel 97, 2000, or 2002 Workbook (default)--> Next-->

c. Browse to find and select the file psa.xls on your desktop. Click Open. Click OK. 

d. Under “what table do you want to import?” leave psa selected-->Next-->

e. Under “Choose the SAS destination” scroll to pick the work library; then, under member, type: psa to name the dataset work.psa. --->Finish
3. Run an exploratory logistic regression model to predict capsule and ask for 90% confidence intervals for the Odds Ratios as follows:

proc logistic data=work.psa;

model capsule (event=”1”) = psa age vol race gleason /risklimits alpha=.10;

4. Examine the output:

1. TESTS OF GLOBAL MODEL FIT

The likelihood ratio test is a global test of fit. The null hypothesis is that none of the predictor variables are related to the outcome (ALL the betas=0). If the likelihood ratio test has a significant p-value, this means that at least one of the predictor variables is significantly related to the outcome (beta not equal to 0). 

More details (to be discussed in class next Monday as well!):

The likelihood ratio test comes directly from the likelihood equation in Maximum Likelihood Estimation. 

When the model is fit with only the intercept (no predictors), the value of the likelihood equation (the probability of our data) at its maximum value is 9.9090187 × 10-111, which translates to a 

-2LogLikelihood (-2LogL) of 506.587. When the model is fit with the intercept and the 5 predictors, the value of the likelihood equation at its maximum value is 6.08546469 × 10-87, which translates to a -2LogLikelihood of 397.

If you subtract the -2LogL of a reduced model (intercept only) from the -2LogL of a full model (intercept+ K predictors), this has a chi-square distribution with K degrees of freedom under the null hypothesis (ALL Betas=0). Here we get a value of 109.5 for a chi-square with 5 degrees of freedom (highly significant, so reject the null!).

If the null hypothesis rejects, this means that at least one of the K predictors is important (at least one of the betas is not equal to 0). Something in our model is predictive!
                                      Model Fit Statistics

                                                          Intercept

                                           Intercept            and

                             Criterion          Only     Covariates


                             AIC             508.587        409.038

                             SC              512.517        432.616

                             -2 Log L        506.587        397.038
               

                                     The LOGISTIC Procedure

                            Testing Global Null Hypothesis: BETA=0

                    Test                 Chi-Square       DF     Pr > ChiSq


                    Likelihood Ratio       109.5494        5         <.0001

                    Score                   90.3463        5         <.0001

                    Wald                    68.0613        5         <.0001

2. PARAMETER ESTIMATES
Analysis of Maximum Likelihood Estimates

                                             Standard          Wald

              Parameter    DF    Estimate       Error    Chi-Square    Pr > ChiSq
              Intercept     1     -5.9887      1.5222       15.4783        <.0001

              psa           1      0.0279     0.00933        8.9134        0.0028

              age           1     -0.0212      0.0193        1.2158        0.2702

              vol           1     -0.0129     0.00759        2.9064        0.0882

              race          1     -0.4351      0.4494        0.9375        0.3329

              gleason       1      1.0541      0.1627       41.9523        <.0001




Odds Ratio Estimates

                                        Point          90% Wald
                          Effect     Estimate      Confidence Limits
                          psa           1.028       1.013       1.044

                          age           0.979       0.948       1.011

                          vol           0.987       0.975       1.000

                          race          0.647       0.309       1.355

                          gleason       2.869       2.196       3.750


                 
5.  Use the “units” statement to calculate the odds of capsule for different units of the continuous predictors.
proc logistic data=work.psa;


model capsule (event="1") = psa age vol race gleason /risklimits;


units psa=10 age=10 ; 

run;
Scroll to the bottom of your output to find:

                       Wald Confidence Interval for Adjusted Odds Ratios

                  Effect          Unit     Estimate     95% Confidence Limits

                  psa          10.0000        1.321        1.100        1.587

                  age          10.0000        0.809        0.554        1.180


6. Sometimes, you might want to get odds ratios by quartile of a continuous variable, particularly if you don’t think the increase in risk is linear across quartiles.  For example, you could look at increasing quartiles of psa level.



proc sort data=work.psa;

by psa; 

proc rank data=work.psa out=work.psa groups=4;

ranks psaq;

var psa;

run;

proc logistic data = work.psa;

        class psaq (ref="0");

        model capsule (event="1") = psaq age gleason vol race;

run;                       


The LOGISTIC Procedure

                                      Odds Ratio Estimates

                                            Point          95% Wald

                       Effect            Estimate      Confidence Limits

                       psaq    1 vs 0       1.621       0.814       3.231

                       psaq    2 vs 0       1.150       0.566       2.338

                       psaq    3 vs 0       2.770       1.315       5.832
                       age                  0.980       0.944       1.018

                       gleason              3.019       2.189       4.163

                       vol                  0.987       0.972       1.002

                   race                 0.733       0.310       1.734

7. The logistic model assumes that continuous predictors are “linear in the logit”, e.g., for the predictor psa the model is: 

Thus, for every 1-unit increase in psa, there should be a linear increase in the logit of capsule, across all levels of psa.  

Fortunately, Ray Balise has written a macro that will plot a predictor variable against the logit of the outcome, so we can easily test this assumption. I modified his original MACRO to make the code a little simpler.
a. Download the macro from the course website (if you haven’t already done so): www.stanford.edu/~kcobb/courses/hrp261-->Logit Plot Macro
b. Open the macro
c. Back in SAS, highlight the macro code and hit the running man icon to run the macro.  

d. Invoke the macro by running the following code:

%logitplot(work.psa, psa, capsule, NumBins = 20);

(The macro with these specifications does the following:

1. ranks observations (1-380) by psa value

2. divides the observations into 20 bins based on their psa values (380/20=19 per bin)
3. Calculates the log odds of capsule=1 in each group as log (#capsule/(19-#capsule))

4. Plots the logit values for the 20 groups against the mean psa level for each group.)

RESULTING GRAPH:

e. Try other bin sizes and see if the overall conclusion changes.

f.  Invoke the macro for the categorical variable psaq by typing in the following:

%logitplot(work.psa, psaq, capsule, NumBins= 4);

8. Next, get SAS to calculate predicted probabilities for you, and put them into a new dataset. Use a model with just psa and gleason.

proc logistic data = work.psa;

        model capsule (event="1") = psa gleason;

output out = outdata l = Lower p = Predicted u = Upper;

run;


9.   Use the interactive data screens to view your new dataset (which we’ve named work.outdata):

a. From the menu select: Solutions(Analysis(Interactive Data Analysis
b. Double click to open: library “work”, dataset “outdata”

c. Scroll right to see the variables that have been added to your data.

10. Print out the predicted values, confidence limits, and residuals.

proc print data=outdata;

var psa gleason capsule lower predicted upper;

run;




FROM PROC PRINT:
       Obs     psa    gleason    capsule     Lower     Predicted     Upper     residuals

         1    0.30       6          0       0.16598     0.21836     0.28170     -0.70195

         2    0.40       5          0       0.05465     0.08852     0.14026     -0.43054

         3    0.50       0          0       0.00007     0.00049     0.00353     -0.03123

         4    0.70       5          0       0.05511     0.08916     0.14111     -0.43218

         5    0.70       5          0       0.05511     0.08916     0.14111     -0.43218
         6    0.80       6          0       0.16850     0.22065     0.28343     -0.70611

         7    1.00       6          0       0.16952     0.22157     0.28414     -0.70778

         8    1.00       5          0       0.05558     0.08982     0.14196     -0.43384

         9    1.00       6          0       0.16952     0.22157     0.28414     -0.70778                   


11. Get predicted probabilities for values of psa&gleason other than those found in the dataset (other than those of the 380 men in the data set).  Beware of making predictions way beyond the values of  psa&gleason observed in your dataset, though.


proc logistic data = psa outmodel = MyEquation;

        model capsule (event = "1") = psa gleason;

run;

data newValues;

input psa gleason;

datalines;

100 10

run;

proc logistic inModel = MyEquation;

        score data = newValues out = NewPredictions;

run;

proc print data=NewPredictions;

var psa gleason p_1;

run;

   Obs    psa    gleason      P_1

                                 1     100       10      0.99640
12. To test for confounders in multivariate regression, try the model with and without the potential confounder, and see if the beta coefficient for the main predictor changes “substantially” (might use 10% as a rule of thumb).

For example, test whether or not race is an important confounder of the relationship between psa and capsule.  Notice that race does not appear to be significantly related to capsule. Is it still a confounder?
proc logistic data = work.psa;


model capsule (event="1") = psa;

run;

proc logistic data = work.psa;

        model capsule (event="1") = psa race;

run;

                           Analysis of Maximum Likelihood Estimates

                                             Standard          Wald

              Parameter    DF    Estimate       Error    Chi-Square    Pr > ChiSq

              Intercept     1     -1.1137      0.1616       47.5168        <.0001

              psa           1      0.0502     0.00925       29.4230        <.0001

                                             Standard          Wald

              Parameter    DF    Estimate       Error    Chi-Square    Pr > ChiSq

              Intercept     1     -1.0781      0.1622       44.1654        <.0001

              psa           1      0.0512     0.00949       29.0371        <.0001

              race          1     -0.5788      0.4187        1.9111        0.1668

13. To test for effect modifiers (interaction), add an interaction term to the model or stratify by the potential effect modifier and compare the estimates of effect in each stratum.

For example, is race an effect modifier of the relationship between psa and capsule?

proc logistic data = work.psa;


model capsule (event="1") = psa race psa*race;

run;

   Analysis of Maximum Likelihood Estimates

                                             Standard          Wald

              Parameter    DF    Estimate       Error    Chi-Square    Pr > ChiSq

              Intercept     1     -1.1904      0.1793       44.0820        <.0001

              psa           1      0.0608      0.0117       26.9250        <.0001

              race          1      0.0954      0.5421        0.0310        0.8603

              psa*race      1     -0.0349      0.0193        3.2822        0.0700



This indicates that high psa levels are more worrisome in white men, though black men overall have a higher risk of capsule=1.

14. The other way to test for interaction is to stratify by race, and look at the ORs/beta coefficients for psa within each stratum. MAKE SURE THE DATASET IS CLOSED BEFORE YOU TRY TO SORT IT.

proc sort data=work.psa; by race; run;

proc logistic data=work.psa;


model capsule (event="1") = psa /risklimits;


by race;

run;

race=0

                                     The LOGISTIC Procedure

                           Analysis of Maximum Likelihood Estimates

                                             Standard          Wald

              Parameter    DF    Estimate       Error    Chi-Square    Pr > ChiSq

              Intercept     1     -1.1904      0.1793       44.0820        <.0001

              psa           1      0.0608      0.0117       26.9250        <.0001
race=1

                        The SAS System      11:15 Tuesday, February 1, 2005  88

-------------------------------------------- race=1 --------------------------------------------

                                     The LOGISTIC Procedure

                           Analysis of Maximum Likelihood Estimates

                                             Standard          Wald

              Parameter    DF    Estimate       Error    Chi-Square    Pr > ChiSq

              Intercept     1     -1.0950      0.5116        4.5812        0.0323

              psa           1      0.0259      0.0153        2.8570        0.0910
 

APPENDIX A:  PROC GPLOT 

Options for plotting symbols in SAS/GRAPH:

Syntax examples:

symbol1 v=star c=yellow h=1 w=1;

symbol2 value=& color=green h=2 w=2;

Options for plotting lines in SAS/GRAPH:

Syntax examples (place within a symbol statement):  

symbol3 v=none c=black w=2 i=join line=5;

Controlled for other factors, being in the top quartile of psa level strongly predicts capsule penetration; but there’s not much difference between being in the 2nd vs. 3rd quartiles, and these two quartiles have only slightly elevated risk compared with the 1st.








This tells SAS to calculate odds ratios comparing each quartile with the bottom quartile (the lowest value is used as the default if you don’t specify).





**PROC RANK will not work unless you use PROC SORT first to order your data by the variable you want to rank.


Dataset must be closed!!





To manually incorporate the units, multiply the estimated beta coefficient for 1-unit by the number of units of interest; then exponentiate to get the corresponding OR:


� EMBED Equation.3  ���





Interpretation: for every 10-unit increase in psa level, there is an estimated 32.1% increase in odds of capsule penetration.





Note that, because we did not specify an alpha level, alpha of .05 was assumed. 


























Interpretation: adjusted for age, volume, race, and gleason score, every 1-unit increase in psa level is associated with a 2.8% increase in the odds of tumor penetration into the capsule.  


Of course, psa level ranges from 0-140 in our dataset, so a 1-unit increase is not a big change.











The risklimits option asks for confidence limits for your odds ratios.


Use the alpha option to generate confidence intervals other than 95% (which is the default).  For example, if you want a 90% CI, specify alpha=.10.




















3. Use the equation/model to get predictions on new observations and output them to a data set (that we’re calling NewPredictions here). 


SAS calls predicted prob. variable P_1 for the predicted probability of capsule=1.


 SAS calls predicted prob. variable P_0 for the predicted probability of capsule=0.





If you do not specify that you want to model the odds of capsule=1, SAS will use numerical or alphabetical order (as in PROC FREQ) and will model the odds of NOT having capsule (capsule=0).  You will get the reciprocal of the OR that you are interested in.


The other way to get around this is to use the descending option in the first line as follows:


proc logistic descending data=psa;


model capsule = psa age vol race gleason /risklimits


run;











2. Make a dataset (we’re calling it newValues here) containing the new values (as many as you want).





1. Use the outmodel option to store the logistic regression model (we’re calling the model MyEquation here).





SAS will not give you ORs if you have an interaction term—because there is not a single OR; but you can calculate the appropriate ORs (for different levels of the predictors) as follows:


� EMBED Equation.3  ���


� EMBED Equation.3  ���





Example of calculating the predicted:


� EMBED Equation.3  ���





Don’t have a scientific calculator handy?  Use SAS!





data _null_;


power=-7.6393 + 0.027*.7 + 1.0593*5;


prob=exp(power)/(1+exp(power));


put power;


put prob;


run;





IN SAS LOG:


-2.3239


0.0891628215





Predicted value and residual are the same for observations 4 and 5, since both men had same psa, gleason, and capsule.





out=newdataset : specifies the name of the new dataset that will contain the ranked variable (here we are writing over our old dataset—always precarious!)


groups specifies how many ranks you want (will run from 0 to #groups-1)


ranks statement names a new variable that will contain the ranks—if you omit the ranks statement SAS will write over the old variable with the ranks, and you will lose the original values.


var statement specifies which variable you want ranked. 





Interpretation: for every 1-unit increase in gleason score, there is an estimated 186.9% increase in the odds of capsule penetration (adjusted for the other things in the model).




















Shows just what the ORs  for quartiles showed: not a linear increase in logit as you increase quartiles, but a steep jump from the 3rd to the 4th quartile.





Nearly significant interaction term indicates some interaction present.





Very little change in beta coefficient for psa; indicates race not an important confounder.




















By designating psaq as a CLASS variable (categorical), SAS automatically dummy codes to represent the four classes of psa quartile.





residual=capsule-predicted 





Predicted probability of capsule 





FINAL MODEL:


Predicted logit (capsule) = -7.6393 + 0.027 (psa) + 1.0593 (gleason)     








In SAS, most PROCs for multivariate regression (GLM, REG, LOGISTIC, PHREG, etc.) allow an OUTPUT statement. Here, you ask SAS to create to a new dataset (out=NewDataSet) which contains all of your original data plus predicted values and confidence limits for predicted values, etc.  


In PROC LOGISTIC, SAS recognizes l, p, u—you just need to name the variables you want.


























Only psa, gleason, and volume are significant at the .10 level. 





To get, for example, the OR and 90% CI for psa: � EMBED Equation.3  ���











The fitted model is: � EMBED Equation.3  ���








These are the p-values for each predictor (from the Wald test) from the null hypothesis that the particular beta coefficient is 0. 





These are tests of whether any of the beta’s are significantly different than zero (if any of them are useful predictors of capsule).








Under the null hypothesis of no difference, this likelihood ratio statistic has a chi-square distribution with 5 degrees of freedom. A value of 109.549 is highly significant.








Likelihood Ratio=


-2LogL(Reduced)- -2LogL(Full) = 506.587-397.038 = 109.549

